// File: Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) private pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: ECDA.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 *
 * @dev Copy of the Zeppelin's library:
 *      https://github.com/OpenZeppelin/openzeppelin-contracts/blob/b0cf6fbb7a70f31527f36579ad644e1cf12fdf4e/contracts/utils/cryptography/ECDSA.sol
 */
library ECDSA {
	/**
	 * @dev Returns the address that signed a hashed message (`hash`) with
	 * `signature`. This address can then be used for verification purposes.
	 *
	 * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
	 * this function rejects them by requiring the `s` value to be in the lower
	 * half order, and the `v` value to be either 27 or 28.
	 *
	 * IMPORTANT: `hash` _must_ be the result of a hash operation for the
	 * verification to be secure: it is possible to craft signatures that
	 * recover to arbitrary addresses for non-hashed data. A safe way to ensure
	 * this is by receiving a hash of the original message (which may otherwise
	 * be too long), and then calling {toEthSignedMessageHash} on it.
	 *
	 * Documentation for signature generation:
	 * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
	 * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
	 */
	function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
		// Divide the signature in r, s and v variables
		bytes32 r;
		bytes32 s;
		uint8 v;

		// Check the signature length
		// - case 65: r,s,v signature (standard)
		// - case 64: r,vs signature (cf https://eips.ethereum.org/EIPS/eip-2098) _Available since v4.1._
		if (signature.length == 65) {
			// ecrecover takes the signature parameters, and the only way to get them
			// currently is to use assembly.
			assembly {
				r := mload(add(signature, 0x20))
				s := mload(add(signature, 0x40))
				v := byte(0, mload(add(signature, 0x60)))
			}
		}
		else if (signature.length == 64) {
			// ecrecover takes the signature parameters, and the only way to get them
			// currently is to use assembly.
			assembly {
				let vs := mload(add(signature, 0x40))
				r := mload(add(signature, 0x20))
				s := and(vs, 0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff)
				v := add(shr(255, vs), 27)
			}
		}
		else {
			revert("invalid signature length");
		}

		return recover(hash, v, r, s);
	}

	/**
	 * @dev Overload of {ECDSA-recover} that receives the `v`,
	 * `r` and `s` signature fields separately.
	 */
	function recover(
		bytes32 hash,
		uint8 v,
		bytes32 r,
		bytes32 s
	) internal pure returns (address) {
		// EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
		// unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
		// the valid range for s in (281): 0 < s < secp256k1n ÷ 2 + 1, and for v in (282): v ∈ {27, 28}. Most
		// signatures from current libraries generate a unique signature with an s-value in the lower half order.
		//
		// If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
		// with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
		// vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
		// these malleable signatures as well.
		require(
			uint256(s) <= 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0,
			"invalid signature 's' value"
		);
		require(v == 27 || v == 28, "invalid signature 'v' value");

		// If the signature is valid (and not malleable), return the signer address
		address signer = ecrecover(hash, v, r, s);
		require(signer != address(0), "invalid signature");

		return signer;
	}

	/**
	 * @dev Returns an Ethereum Signed Message, created from a `hash`. This
	 * produces hash corresponding to the one signed with the
	 * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
	 * JSON-RPC method as part of EIP-191.
	 *
	 * See {recover}.
	 */
	function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
		// 32 is the length in bytes of hash,
		// enforced by the type signature above
		return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
	}

	/**
	 * @dev Returns an Ethereum Signed Typed Data, created from a
	 * `domainSeparator` and a `structHash`. This produces hash corresponding
	 * to the one signed with the
	 * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
	 * JSON-RPC method as part of EIP-712.
	 *
	 * See {recover}.
	 */
	function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
		return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
	}
}

// File: ERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: ERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./IERC721Metadata.sol";
import "./Address.sol";
import "./Context.sol";
import "./Strings.sol";
import "./ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
abstract contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Base URI for token metadata
    string public URI;
    
    // Record sell data of token
    struct Sell {
        uint256 price;
        bool forSell;
        uint8 sellType;     // option 0 = fixed price, 1 = auction
        uint64 endTime;
    }
    
    // Mapping from token Id to Sell 
    mapping(uint256 => Sell) public sellData;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return URI;
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver(to).onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {

        require(!sellData[tokenId].forSell, "HomeJab: Ongoing sell");

        require(from != to, "HomeJab: Invalid input");

    }
    
}

// File: ERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./ERC721.sol";
import "./IERC721Enumerable.sol";

/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual override {
        super._beforeTokenTransfer(from, to, tokenId);

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}

// File: HomeJab.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./ERC721Enumerable.sol";
import "./ECDA.sol";

// Link to Payment token
interface IBEP20TOKEN {

    // Transfer Payment token
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns(bool);
    
    // Return spending allowance
    function allowance(
      address _owner,
      address _spender
    ) external returns (uint256);

}

/**
 * @title HomeJab Marketplace version 1.0
 *
 * @author HomeJab(Joe Jesuele)
 */
contract HomeJab is ERC721Enumerable {

    // Address of owner of the contract
    address public owner;

    // Marketplace fee in terms of percentage
    uint8 public marketplaceFee;

    // Maximum batch size allowed
    uint8 public batchSize;

    // Contract address of payment token
    address public paymentToken;

    /**
	 * @notice EIP-712 contract's domain typeHash,
	 *      see https://eips.ethereum.org/EIPS/eip-712#rationale-for-typehash
	 *
	 * @dev Note: we do not include version into the domain typehash/separator,
	 *      it is implied version is concatenated to the name field, like "HomeJab"
	 */
	// keccak256("EIP712Domain(string name,uint256 chainId,address verifyingContract)")
	bytes32 public constant DOMAIN_TYPEHASH = 0x8cad95687ba82c2ce50e74f7b754645e5117c3a5bec8151c0726d5857980a866;

	/**
	 * @notice EIP-712 contract's domain separator,
	 *      see https://eips.ethereum.org/EIPS/eip-712#definition-of-domainseparator
	 */
	bytes32 public immutable DOMAIN_SEPARATOR;

	/**
	 * @notice EIP-712 permit (EIP-2612) struct typeHash,
	 *      see https://eips.ethereum.org/EIPS/eip-712#rationale-for-typehash
	 */
	// keccak256("Permit(address owner,address operator,uint256 tokenId,uint256 nonce,uint256 deadline)")
	bytes32 public constant PERMIT_TYPEHASH = 0xee2282d7affd5a432b221a559e429129347b0c19a3f102179a5fb1859eef3d29;
    
    /**
	 * @dev A record of nonces for signing/validating signatures in EIP-712 based
	 *      `permit` and `permitForAll` functions
	 *
	 * @dev Each time the nonce is used, it is increased by one, meaning reordering
	 *      of the EIP-712 transactions is not possible
	 *
	 * @dev Inspired by EIP-2612 extension for ERC20 token standard
	 *
	 * @dev Maps token owner address => token owner nonce
	 */
	mapping(address => uint256) public permitNonces;

    // Record bidder address and bidding amount
    struct Bid {
        address from;
        uint256 amount;
    }

    // Mapping from token Id to address of creator 
    mapping(uint256 => address) public creator;

    // Mapping from token Id to royalty fee associated with
    mapping(uint256 => uint8) public royalty;

    // Mapping from token Id to user bids
    mapping(uint256 => Bid[]) public auctionBids;

    /**
	 * @dev Fired in transferOwnership() when ownership transfered 
	 *
	 * @param _previousOwner an address of previous owner
	 * @param _newOwner an address of new owner
	 */
    event OwnershipTransferred(
        address indexed _previousOwner,
        address indexed _newOwner
    );

    /**
	 * @dev Fired in mint() when NFT minted
	 *
	 * @param _to an address of NFT minted to
	 * @param _id token Id of minted NFT 
     * @param _royaltyFee fees charged as royalty in terms of percentage
	 */
    event Minted(
        address indexed _to,
        uint256 indexed _id,
        uint8 _royaltyFee
    );

    /**
	 * @dev Fired in listForSell() when NFT listed for sell 
	 *
	 * @param _by an address of NFT owner
	 * @param _id token Id of listed NFT 
     * @param _price sell price for the NFT
     * @param _type sell type (0 = fixed price, 1 = english auction)
	 */
    event Listed(
        address indexed _by,
        uint256 indexed _id,
        uint256 _price,
        uint8 _type
    );

    /**
	 * @dev Fired in revokeSell() when NFT delisted  
	 *
	 * @param _by an address of NFT owner
	 * @param _id token Id of minted NFT
     */
    event Delisted(
        address indexed _by,
        uint256 indexed _id
    );

    /**
	 * @dev Fired in editPrice() when NFT relisted at new price 
	 *
	 * @param _by an address of NFT owner
	 * @param _id token Id of relisted NFT
     * @param _oldPrice sell price before relisting
     * @param _newPrice sell price after relisting
	 */
    event Relisted(
        address indexed _by,
        uint256 indexed _id,
        uint256 _oldPrice,
        uint256 _newPrice
    );

    /**
	 * @dev Fired in buy() when NFT bought 
	 *
     * @param _from an address of seller
	 * @param _to an address of buyer
	 * @param _id token Id of sold NFT 
     * @param _price price paid by buyer 
     * @param _marketplaceFee fees charged by marketplace 
     * @param _royaltyFee fees charged as royalty
	 */
    event Bought(
        address indexed _from,
        address indexed _to,
        uint256 indexed _id,
        uint256 _price,
        uint256 _marketplaceFee,
        uint256 _royaltyFee
    );

    /**
	 * @dev Fired in bidOnAuction() and revokeBid() when bid is placed/revoked
	 *      for english auction successfully
	 *
	 * @param _from an address of bidder
	 * @param _id token Id of bidded NFT
	 * @param _amount defines bid amount to be paid (zero if bid is revoked)
	 */
    event BidUpdate(
        address indexed _from,
        uint256 indexed _id,
        uint256 _amount
    );

    // To check ownership for restricted access functions
    modifier onlyOwner {

        verifyOwner();
    
        _;

    }

    // To check batch request lies within limits
    modifier batchLimit(uint256 length_) {

        verifyBatchLimit(length_);

        _;

    }

    /**
	 * @dev Creates/deploys HomeJab Marketplace
	 *
	 * @param fee_ marketplace fee in terms of percentage
	 * @param batchSize_ maximum allowed amount in a batch
	 * @param baseURI_ base URL for token metadata end point
	 * @param paymentToken_ address of BEP20 token for payments
	 */
    constructor(
        uint8 fee_,
        uint8 batchSize_,
        string memory baseURI_,
        address paymentToken_
    ) ERC721("HomeJab Market","REAL") {
        
        require(
            fee_ <= 10 && paymentToken_ != address(0),
            "HomeJab: Invalid input"
        );
        
        //----------Set Internal State----------//
        owner = msg.sender;

        marketplaceFee = fee_;

        batchSize = batchSize_;

        URI = baseURI_;

        paymentToken = paymentToken_;

        // build the EIP-712 contract domain separator, see https://eips.ethereum.org/EIPS/eip-712#definition-of-domainseparator
		// note: we specify contract version in its name
		DOMAIN_SEPARATOR = keccak256(abi.encode(DOMAIN_TYPEHASH, keccak256(bytes("HomeJab")), block.chainid, address(this)));

    }

    /**
     * @dev Returns address of the owner
     */
    function getOwner() external view returns(address) {
        return owner;
    }    

    /**
     * @dev Returns royalty info as par the NFT Royalty Standard(EIP-2981) 
     * 
     * @notice Called with the sale price to determine how much royalty
     *          is owed and to whom.
     * @param id_ the NFT asset queried for royalty information
     * @param price_ the sale price of the NFT asset specified by id_
     * @return receiver_ address of who should be sent the royalty payment
     * @return royaltyAmount_ the royalty payment amount for price_
     */
    function royaltyInfo(uint256 id_, uint256 price_)
        external
        view
        returns (address receiver_, uint256 royaltyAmount_)
    {
    
        receiver_ = creator[id_];
        
        royaltyAmount_ = (price_ * royalty[id_]) / 100;

    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner_`).
     *      Can only be called by the current owner.
     * 
     * @param newOwner_ address of new owner
     */
    function transferOwnership(address newOwner_) external onlyOwner {
        
        require(
            newOwner_ != address(0),
            "HomeJab: new owner is the zero address"
        );

        emit OwnershipTransferred(owner, newOwner_);
        
        owner = newOwner_;
        
    }

    /**
     * @dev Sets marketplace fee 
     *
     * @notice Restricted access function, should be called by owner only 
     * @param fee_ defines marketplace fee in terms of percentage
     */
    function setMarketplaceFee(uint8 fee_) 
        external 
        onlyOwner
    {
        
        require(fee_ <= 10, "HomeJab: Invalid input");
        
        marketplaceFee = fee_;

    }

    /**
     * @dev Sets batch size for batch operations 
     *
     * @notice Restricted access function, should be called by owner only 
     * @param batchSize_ defines batch size
     */
    function setBatchSize(uint8 batchSize_) 
        external 
        onlyOwner
    {    
        batchSize = batchSize_;
    }

    /**
     * @dev Sets base URI for token metadata 
     *
     * @notice Restricted access function, should be called by owner only 
     * @param baseURI_ defines base URI
     */
    function setBaseURI(string memory baseURI_) 
        external 
        onlyOwner
    {
        URI = baseURI_;
    }

    /**
     * @dev Sets payment token interface 
     *
     * @notice Restricted access function, should be called by owner only 
     * @param paymentToken_ defines contract address of BEP20 token
     */
    function setPaymentInterface(address paymentToken_) 
        external 
        onlyOwner
    {
        
        require(paymentToken_ != address(0), "HomeJab: Invalid input");

        paymentToken = paymentToken_;
    
    }

    /**
     * @dev Mint new NFT
     *
     * @param royalty_ defines royalty fee in terms of percentage
     * @return id_ defines token id that is minted for creator
     */
    function mint(uint8 royalty_) public returns(uint256 id_) {

        require(royalty_ <= 10, "HomeJab: Invalid royalty fee");

        // Get Id
        uint256 _id = totalSupply() + 1;

        // Record creator address
        creator[_id] = msg.sender;

        // Record royalty fee
        royalty[_id] = royalty_;

        // Mint token with given Id
        _safeMint(msg.sender, _id);

        // Emit an event
        emit Minted(msg.sender, _id, royalty_);

        return _id;

    }

    /**
     * @dev Mints new NFTs in batch
     *
     * @param amount_ defines amount of NFTs to be minted for creator
     * @param royalty_ defines royalty fee per NFT in terms of percentage
     */
    function mintBatch(uint8 amount_, uint8 royalty_)
        external
        batchLimit(amount_)  
    {

        for(uint8 i; i < amount_; i++) {
            
            mint(royalty_);

        }

    }

    /**
     * @dev Mints new NFT to creator and transfer it to buyer
     *
     * @param creator_ defines address creator
     * @param royalty_ defines royalty fee per NFT in terms of percentage
     * @param price_ defines price of given NFT 
     */
    function lazyMint(address creator_, uint8 royalty_, uint256 price_) public {
        
        require(royalty_ <= 10, "HomeJab: Invalid royalty fee");

        // Get Id
        uint256 _id = totalSupply() + 1;

        // Record creator address
        creator[_id] = creator_;

        // Record royalty fee
        royalty[_id] = royalty_;

        // Mint token with given Id
        _safeMint(creator_, _id);

        // Emit an event
        emit Minted(creator_, _id, royalty_);
        
        // Calculates marketplace fee
        uint256 _marketplaceFee = (price_ * marketplaceFee) / 100; 
        
        //------------------ Transfer price and fees ------------------//
        IBEP20TOKEN(paymentToken).transferFrom(msg.sender, owner, _marketplaceFee);

        IBEP20TOKEN(paymentToken).transferFrom(msg.sender, creator_, price_ - _marketplaceFee);
        
        // Transfer NFT to buyer
        _transfer(creator_, msg.sender, _id);

    }

    /**
     * @dev List NFT for sell at fixed price
     *
     * @param id_ defines token Id to be listed for sell
     * @param price_ defines desired sell price of NFT
     */
    function listForSell(uint256 id_, uint256 price_)
        public
    {

        require(msg.sender == ownerOf(id_), "HomeJab: only NFT owner can list NFT for sell");

        require(!sellData[id_].forSell, "HomeJab: Sell is active");

        // Record sell data
        sellData[id_] = Sell(price_, true, 0, 0);

        // Emit an event
        emit Listed(msg.sender, id_, price_, 0);

    }

    /**
     * @dev List NFTs for sell at fixed price
     *
     * @param id_ defines array of token Ids to be listed for sell
     * @param price_ defines desired sell price per NFT
     */
    function listForSellBatch(uint256[] memory id_, uint256 price_)
        external
        batchLimit(id_.length)
    {
        
        for(uint8 i; i < id_.length; i++) {

            listForSell(id_[i], price_);

        }

    }

    /**
     * @dev Edit sell price for given NFT
     *
     * @param id_ defines token Id of NFT
     * @param newPrice_ defines new price to be set for sell
     */
    function editPrice(uint256 id_, uint256 newPrice_) 
        external
    {
        require(msg.sender == ownerOf(id_), "HomeJab: only NFT owner can edit price");

        require(
            sellData[id_].forSell && sellData[id_].sellType == 0,
            "HomeJab: Fixed price sell is not active"
        );
        
        // Get old price
        uint256 _oldPrice = sellData[id_].price;
        
        // Record new price
        sellData[id_].price = newPrice_;

        // Emit an event
        emit Relisted(msg.sender, id_, _oldPrice, newPrice_);

    }    

    /**
     * @dev Buy given NFT
     *
     * @param id_ defines token Id of NFT
     */
    function buy(uint256 id_)
        external
    {
        
        require(sellData[id_].forSell, "HomeJab: Sell is not active");

        // Get address of seller
        address _seller = ownerOf(id_);

        // Get sell price of given NFT
        uint256 _price = sellData[id_].price;

        // Calculate marketplace fee
        uint256 _marketplaceFee = (_price * marketplaceFee) / 100;
        
        // Calculate royalty fee
        uint256 _royaltyFee = (_price * royalty[id_]) / 100;

        //------------------ Transfer price and fees ------------------//    
        IBEP20TOKEN(paymentToken).transferFrom(
            msg.sender,
            _seller,
            _price - (_marketplaceFee + _royaltyFee)
        );

        IBEP20TOKEN(paymentToken).transferFrom(msg.sender, owner, _marketplaceFee);
        
        IBEP20TOKEN(paymentToken).transferFrom(msg.sender, creator[id_], _royaltyFee);

        // Remove sell data
        sellData[id_] = Sell(0, false, 0, 0);

        // Transfer NFT to buyer
        _transfer(_seller, msg.sender, id_);

        // Emit an event
        emit Bought(_seller, msg.sender, id_, _price, _marketplaceFee, _royaltyFee);

    }

    /**
     * @dev List NFT for sell on english auction
     *
     * @param id_ defines token Id to be listed for sell
     * @param basePrice_ defines desired base price of NFT
     * @param endTime_ defines auction ending time in timestamp
     */
    function listForEnglishAuction(
        uint256 id_,
        uint256 basePrice_,
        uint64 endTime_
    ) public {
        
        require(msg.sender == ownerOf(id_), "HomeJab: only NFT owner can list NFT for sell");

        require(!sellData[id_].forSell, "HomeJab: Sell is active");

        require(endTime_ > block.timestamp, "HomeJab: Past time");

        // Record sell data
        sellData[id_] = Sell(basePrice_, true, 1, endTime_);

        // Emit an event
        emit Listed(msg.sender, id_, basePrice_, 1);

    }
    
    /**
     * @dev List NFTs for sell on english auction
     *
     * @param id_ defines array of token Ids to be listed for sell
     * @param basePrice_ defines desired base price per NFT
     * @param endTime_ defines auction ending time in timestamp
     */
    function listForEnglishAuctionBatch(
        uint256[] memory id_,
        uint256 basePrice_,
        uint64 endTime_
    )
        external
        batchLimit(id_.length)
    {

        for(uint8 i; i < id_.length; i++) {

            listForEnglishAuction(id_[i], basePrice_, endTime_);

        }

    }

    /** 
     * @dev returns bid count of NFT listed in english auction
     * 
     * @param id_ unsigned integer defines token Id 
     */
    function getBidCount(uint256 id_) public view returns (uint256) {
      return auctionBids[id_].length;
    }

    /** 
     * @dev Bid for NFT listed in english auction
     *
     * @param id_ defines token Id of auctioned NFT 
     * @param bidPrice_ defines number of payment tokens offered for bid  
     */
    function bidOnAuction(
        uint256 id_,
        uint256 bidPrice_
    ) external {

        require(
            isAuctionActive(id_),
            "HomeJab: Auction is not active"
        );

        // Get last bid amount
        uint256 _lastBid = (getBidCount(id_) == 0) ?
                            sellData[id_].price - 1 : auctionBids[id_][getBidCount(id_) - 1].amount;  

        require(
            bidPrice_ > _lastBid &&
            IBEP20TOKEN(paymentToken).allowance(msg.sender, address(this)) > bidPrice_,
            "HomeJab: Bid price is less than last bid price / base price"
        );

        // Wrap bid information
        Bid memory newBid = Bid(msg.sender, bidPrice_);

        // Record bid information
        auctionBids[id_].push(newBid);
        
        // Emit an event
        emit BidUpdate(msg.sender, id_, bidPrice_);

    }

    /** 
     * @dev Revoke existing Bid for NFT listed in english auction
     *
     * @param id_ defines token Id of auctioned NFT 
     * @param bidIndex_ defines index of bid to be revoked 
     */
    function revokeBid(
        uint256 id_,
        uint256 bidIndex_
    ) external {

        require(
            isAuctionActive(id_),
            "HomeJab: Auction is not active"
        );

        require(
            msg.sender == auctionBids[id_][bidIndex_].from,
            "HomeJab: Invalid bidder"
        );

        // Get last index
        uint256 _lastIndex = getBidCount(id_) - 1;
        
        // Remove given bid index by shifting elements from right to left
        for(uint i = bidIndex_; i < _lastIndex; i++) {
            auctionBids[id_][i] = auctionBids[id_][i + 1];
        }
        
        // Remove last index
        auctionBids[id_].pop();

        // Emit an event
        emit BidUpdate(msg.sender, id_, 0);

    }

    /**
     * @dev Accept bid for NFT listed in english auction 
     *
     * @notice on success NFT is transfered to bidder and seller gets the amount
     * @notice limited access function that can be called by seller only
     * 
     * @param id_ defines token Id of auctioned NFT
     * @param bidIndex_ defines index of given bid to accept for finalize auction
     */
    function acceptBid(
        uint256 id_,
        uint256 bidIndex_
    ) external {

        require(
            sellData[id_].forSell && sellData[id_].sellType == 1,
            "HomeJab: Not listed for auction"
        );

        require(msg.sender == ownerOf(id_), "HomeJab: Not an owner of NFT");

        // Get address of bidder
        address _bidder = auctionBids[id_][bidIndex_].from;

        // Get bid amount of given index
        uint256 _price = auctionBids[id_][bidIndex_].amount;

        // Calculate marketplace fee
        uint256 _marketplaceFee = (_price * marketplaceFee) / 100;
        
        // Calculate royalty fee
        uint256 _royaltyFee = (_price * royalty[id_]) / 100;

        //------------------ Transfer price and fees ------------------//    
        IBEP20TOKEN(paymentToken).transferFrom(
            _bidder,
            msg.sender,
            _price - (_marketplaceFee + _royaltyFee)
        );

        IBEP20TOKEN(paymentToken).transferFrom(_bidder, owner, _marketplaceFee);
        
        IBEP20TOKEN(paymentToken).transferFrom(_bidder, creator[id_], _royaltyFee);

        // Remove sell data
        sellData[id_] = Sell(0, false, 0, 0);

        // Clear Bidding information
        delete auctionBids[id_];

        // Transfer NFT to buyer
        _transfer(msg.sender, _bidder, id_);
        
        // Emit an event
        emit Bought(msg.sender, _bidder, id_, _price, _marketplaceFee, _royaltyFee);
        
    }

    /**
     * @dev Mint a NFT and List it for sell at fixed price/ base price(english auction)
     *
     * @param royalty_ defines royalty fee in terms of percentage
     * @param price_ defines desired sell price/ base price of NFT
     * @param endTime_ defines auction ending time in timestamp(Value must be '0' For fixed price sell)
     */
    function mintAndList(uint8 royalty_, uint256 price_, uint64 endTime_)
        public
    {
        
        // Mint new NFT and record Id
        uint256 _id = mint(royalty_);

        if(endTime_ == 0) {

            // List NFT for fixed price sell
            listForSell(_id, price_);

        } else {
            
            // List NFT for english auction
            listForEnglishAuction(_id, price_, endTime_);

        }

    }

    /**
     * @dev Mint NFTs and List them for sell at fixed price/ base price(english auction)
     *
     * @param amount_ defines amount of NFTs minted and listed
     * @param royalty_ defines royalty fee per NFT in terms of percentage
     * @param price_ defines desired sell price/ base price per NFT
     * @param endTime_ defines auction ending time in timestamp(Value must be '0' For fixed price sell)
     */
    function mintAndListBatch(uint8 amount_, uint8 royalty_, uint256 price_, uint64 endTime_)
        external
        batchLimit(amount_)
    {
        
        for(uint8 i; i < amount_; i++) {

            mintAndList(royalty_, price_, endTime_);

        }

    }

    /**
     * @dev Ends ongoing sell for given NFT
     *
     * @param id_ defines token Id of NFT
     */
    function revokeSell(uint256 id_) 
        external
    {

        require(msg.sender == ownerOf(id_), "HomeJab: only NFT owner can revoke sell");

        require(sellData[id_].forSell, "HomeJab: Sell is not active");

        // Remove sell data
        sellData[id_] = Sell(0, false, 0, 0);

        // Clear Bidding information in case of English auction
        delete auctionBids[id_];
        
        // Emit an event
        emit Delisted(msg.sender, id_);   

    }

    /**
     * @dev Called internally to verify that english auction is running and active
     */
    function isAuctionActive(uint256 id_) internal view returns(bool) {
        return sellData[id_].forSell && sellData[id_].sellType == 1 && sellData[id_].endTime > block.timestamp;
    }
    
    /**
     * @dev Called internally to verify that function is being called by HomeJab owner
     */
    function verifyOwner() internal view {

        require(
            msg.sender == owner,
            "HomeJab: Not an owner"
        );
        
    }
    
    /**
     * @dev Called internally to verify that batch request lies within permissible limit
     */
    function verifyBatchLimit(uint256 length_) internal view {

        require(
            length_ <= batchSize,
            "HomeJab: Invalid batch size"
        );

    }

    // ===== End: ERC721 mutative functions (transfers, approvals) =====

	// ===== Start: Meta-transactions Support =====

	/**
	 * @notice Change or reaffirm the approved address for an NFT on behalf
	 *
	 * @dev Executes approve(_operator, _tokenId) on behalf of the token owner
	 *      who EIP-712 signed the transaction, i.e. as if transaction sender is the EIP712 signer
	 *
	 * @dev Sets the `_tokenId` as the allowance of `_operator` over `_owner` token,
	 *      given `_owner` EIP-712 signed approval
	 *
	 * @dev Emits `Approval` event in the same way as `approve` does
	 *
	 * @dev Requires:
	 *     - `_operator` to be non-zero address
	 *     - `_exp` to be a timestamp in the future
	 *     - `v`, `r` and `s` to be a valid `secp256k1` signature from `_owner`
	 *        over the EIP712-formatted function arguments.
	 *     - the signature to use `_owner` current nonce (see `permitNonces`).
	 *
	 * @dev For more information on the signature format, see the
	 *      https://eips.ethereum.org/EIPS/eip-2612#specification
	 *
	 * @param _owner owner of the token to set approval on behalf of,
	 *      an address which signed the EIP-712 message
	 * @param _operator new approved NFT controller
	 * @param _tokenId token ID to approve
	 * @param _exp signature expiration time (unix timestamp)
	 * @param v the recovery byte of the signature
	 * @param r half of the ECDSA signature pair
	 * @param s half of the ECDSA signature pair
	 */
	function permit(address _owner, address _operator, uint256 _tokenId, uint256 _exp, uint8 v, bytes32 r, bytes32 s) public {
		// derive signer of the EIP712 Permit message, and
		// update the nonce for that particular signer to avoid replay attack!!! ----------->>> ↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓↓
		address signer = __deriveSigner(abi.encode(PERMIT_TYPEHASH, _owner, _operator, _tokenId, permitNonces[_owner]++, _exp), v, r, s);

		// perform message integrity and security validations
		require(signer == _owner, "invalid signature");
		require(block.timestamp < _exp, "signature expired");

		// delegate call to `__approve` - execute the logic required
		_approve(_operator, _tokenId);
	}

    /**
	 * @dev Auxiliary function to verify structured EIP712 message signature and derive its signer
	 *
	 * @param abiEncodedTypehash abi.encode of the message typehash together with all its parameters
	 * @param v the recovery byte of the signature
	 * @param r half of the ECDSA signature pair
	 * @param s half of the ECDSA signature pair
	 */
	function __deriveSigner(bytes memory abiEncodedTypehash, uint8 v, bytes32 r, bytes32 s) private view returns(address) {
		// build the EIP-712 hashStruct of the message
		bytes32 hashStruct = keccak256(abiEncodedTypehash);

		// calculate the EIP-712 digest "\x19\x01" ‖ domainSeparator ‖ hashStruct(message)
		bytes32 digest = keccak256(abi.encodePacked("\x19\x01", DOMAIN_SEPARATOR, hashStruct));

		// recover the address which signed the message with v, r, s
		address signer = ECDSA.recover(digest, v, r, s);

		// return the signer address derived from the signature
		return signer;
	}

}
// File: IERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: IERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: IERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _ALPHABET = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _ALPHABET[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}